



---

# گزارش کار آزمایشگاه DSD

---

آزمایش شماره 10



11 خرداد 1400

عرشیا اخوان  
محمدرضا عیدی  
علیرضا ایلامی

شماره آزمایش: <b>10</b>	موضوع: پیاده سازی یک پردازنده ساده	تاریخ آزمایش: 11 خرداد 1400
عرشیا اخوان <b>97110422</b>	محمد رضا عبدی <b>97110285</b>	علیرضا ایلامی <b>97101286</b>

مقدمه:

یک پردازنده با معماری پشته ای دارای یک پشته با 8 ثبات 8 بیتی است. این پردازنده دارای 8 دستور در مجموعه دستورالعمل های خود است. همچنین دارای حافظه به اندازه 256 خانه 8 بیتی است که 8 خانه آخر آن، به صورت Memory Mapped I/O هستند.

شرح آزمایش:

کلا دو تا استیج داریم: IFIDC و EXEC

در استیج IFIDC، دستور را Fetch میکنیم، آن را Decode میکنیم و در نهایت آن را کنترل میکنیم و به ماژول Exec می‌دهیم تا کار آن دستور را انجام دهد.

در استیج EXEC، به فراخور نوع دستور، از ماژولهای مختلفی استفاده میکنیم که در زیر توضیح داده شده اند: برای دستورهای ADD, SUB از ماژول ALU استفاده میکنیم.

برای دستورهای JUMP, JUMPZ, JUMPS از ماژول PC استفاده میکنیم.

برای دستورهای PUSHC, PUSHM, POPM از ماژول PP استفاده میکنیم.

ابتدا استیج IFIDC:

```
`define INIT      2'b00
`define IF        2'b01
`define ID        2'b10
`define OP_STL    4'b1000
```

```
module IFIDC (clk,
```

```

        en,
        rstn,
        pc,
        IS_ready,
        control_bus,
        data);

parameter INST_CAP = 20;
parameter INST_LEN = 12;
parameter DATA_LEN = 8;

input clk, rstn, en;
input [$clog2(INST_CAP):0] pc;
output reg [3:0] control_bus;
output reg [DATA_LEN-1:0] data;
output reg IS_ready;

reg [INST_LEN-1:0] inst_mem [INST_CAP-1:0];
reg [INST_LEN-1:0] inst;

reg [1:0] state;

always @(posedge clk or negedge rstn) begin
    if (!rstn && !en) begin
        control_bus <= `OP_STL;
        data <= 8'b x;
        state <= `INIT;
        IS_ready <= 0;
    end
    else begin
        case(state)
            `INIT:
                begin
                    if (en) begin
                        state <= `IF;
                    end
                    IS_ready <= 0;
                end
            `IF:
                begin
                    inst <= inst_mem[pc];
                    state <= `ID;
                end
            `ID:
                begin
                    control_bus[3:0] <= inst[INST_LEN-1:DATA_LEN];
                    data <= inst[DATA_LEN-1:0];
                    state <= `INIT;
                    IS_ready <= 1;
                end
        endcase
    end
end

```

```

end

//debugging
always @(*)
    $display($time, "\t [IFIDC::%d] pc = %d control_bus = %b addr_imm = %d IS_ready = %b
rstn = %b en = %b", state, pc, control_bus, data, IS_ready, rstn, en);

endmodule

```

ابتدا در case: INIT هستیم. وقتی  $\text{enable} = 1$  شد وارد فاز IF: Instruction Fetch میشویم. و دستور را از حافظه میخوانیم و اصطلاحاً fetch میکنیم.

مرحله بعدی ID: Instruction Decode است. که در آن بخشهای مختلف دستور را جدا میکنیم.

در پایان مجدداً به فاز INIT برمیگردیم و سیگنال  $\text{is\_ready} = 1$  میکنیم. (برای اینکه ماژول های داخلی EXEC بدانند کار استیج قبلی برای یک دستور تمام شده و می توانند آن را اجرا نمایند).

استیج EXEC:

```

`define NXTI 2'b00
`define EXIT 2'b11

module EXEC (clk,
             en,
             rstn,
             pc,
             control_bus,
             addr_const,
             stk_data_in,
             stk_push,
             stk_pop,
             stk_data_out,
             mem_data_in,
             mem_addr,
             mem_r_en,
             mem_w_en,
             mem_data_out,
             fin_sig);

    parameter DATA_LEN = 8;

```

```

parameter ADDR_LEN = 8;
parameter INST_CAP = 20;

input clk, en, rstn;
input [3:0] control_bus;
input [DATA_LEN-1:0] stk_data_out, mem_data_out, addr_const;
output wire [DATA_LEN-1:0] stk_data_in, mem_data_in;
output wire [ADDR_LEN-1:0] mem_addr;
output wire stk_push, stk_pop, mem_r_en, mem_w_en, fin_sig;
output wire [$clog2(INST_CAP):0] pc;
wire alu_z_flag, alu_s_flag, alu_fin_sig, wbpb_fin_sig, pc_fin_sig;

assign fin_sig = (alu_fin_sig | wbpb_fin_sig | pc_fin_sig);

ALU #(
    .DATA_LEN(DATA_LEN)
) alu0 (
    .clk(clk),
    .rstn(rstn),
    .en(en),
    .control_bus(control_bus),
    .z_flag(alu_z_flag),
    .s_flag(alu_s_flag),
    .stk_data_in(stk_data_in),
    .stk_push(stk_push),
    .stk_pop(stk_pop),
    .stk_data_out(stk_data_out),
    .fin_sig(alu_fin_sig)
);

PP #(
    .DATA_LEN(DATA_LEN),
    .ADDR_LEN(ADDR_LEN)
) pp0 (
    .clk(clk),
    .rstn(rstn),
    .en(en),
    .control_bus(control_bus),
    .addr_const(addr_const),
    .stk_data_in(stk_data_in),
    .stk_push(stk_push),
    .stk_pop(stk_pop),
    .stk_data_out(stk_data_out),
    .mem_data_in(mem_data_in),
    .mem_addr(mem_addr),
    .mem_r_en(mem_r_en),
    .mem_w_en(mem_w_en),
    .mem_data_out(mem_data_out),
    .fin_sig(wbpb_fin_sig)
);

PC #(

```

```

        .DATA_LEN(DATA_LEN),
        .INST_CAP(INST_CAP)
    ) pc0 (
        .clk(clk),
        .rstn(rstn),
        .en(en),
        .control_bus(control_bus),
        .pc(pc),
        .z_flag(alu_z_flag),
        .s_flag(alu_s_flag),
        .stk_pop(stk_pop),
        .stk_push(stk_push),
        .stk_data_out(stk_data_out),
        .fin_sig(pc_fin_sig)
    );

endmodule

```

در ماژول EXEC صرفاً سیم‌کشی‌ها انجام می‌شود و از سه submodule آن، instantiate می‌کنیم.

حال به سراغ ماژول‌های داخلی استیج EXEC می‌رویم:

ماژول ALU:

```

`define INIT      4'b0000
`define OP1_POP   4'b0001
`define OP1_RCV   4'b0010
`define OP1_STR   4'b0011
`define OP2_POP   4'b0100
`define OP2_RCV   4'b0101
`define OP2_STR   4'b0110
`define ALU       4'b0111
`define PUSH      4'b1000
`define PUSH_W    4'b1001

module ALU (control_bus,
            clk,
            en,
            rstn,
            z_flag,
            s_flag,
            stk_data_in,

```

```

        stk_push,
        stk_pop,
        stk_data_out,
        fin_sig);

parameter DATA_LEN = 8;

input clk, rstn, en;
input [3:0] control_bus;
input [DATA_LEN-1:0] stk_data_out;
output reg [DATA_LEN-1:0] stk_data_in;
output reg stk_push, stk_pop;
output reg z_flag, s_flag, fin_sig;

wire asn    = !control_bus[0];
wire alu_en = (en && control_bus[1] && control_bus[2]);

reg [3:0] state;

reg signed [DATA_LEN-1:0] op1, op2, result;

always @(posedge clk, negedge rstn) begin
    if (!rstn && !alu_en) begin
        state      <= `INIT;
        stk_data_in <= {DATA_LEN{1'bz}};
        stk_push    <= 1'bz;
        stk_pop      <= 1'bz;
        s_flag       <= 0;
        z_flag       <= 0;
        fin_sig      <= 0;
    end
    else begin
        case(state)
            `INIT:
                begin
                    if (alu_en) begin
                        state <= `OP1_POP;
                    end
                    fin_sig    <= 0;
                    stk_data_in <= {DATA_LEN{1'bz}};
                    stk_push    <= 1'bz;
                    stk_pop      <= 1'bz;
                end
            `OP1_POP:
                begin
                    stk_push <= 0;
                    stk_pop <= 1;
                    state    <= `OP1_RCV;
                end
            `OP1_RCV:
                begin
                    stk_pop <= 0;

```

```

        state    <= `OP1_STR;
    end
    `OP1_STR:
    begin
        op1      <= stk_data_out;
        state    <= `OP2_POP;
    end
    `OP2_POP:
    begin
        stk_pop  <= 1;
        state    <= `OP2_RCV;
    end
    `OP2_RCV:
    begin
        stk_pop  <= 0;
        state    <= `OP2_STR;
    end
    `OP2_STR:
    begin
        op2      <= stk_data_out;
        state    <= `ALU;
    end
    `ALU:
    begin
        if (asn)
            result <= op2 + op1;
        else
            result <= op2 - op1;
            state <= `PUSH;
        end
    end
    `PUSH:
    begin
        s_flag    <= (result < 0);
        z_flag    <= (result == 0);
        stk_push   <= 1;
        stk_data_in <= result;
        state      <= `PUSH_W;
    end
    `PUSH_W:
    begin
        stk_push   <= 0;
        state      <= `INIT;
        fin_sig    <= 1;
    end
endcase
end
end

//debugging
always @(*)
    $display($time, "\t [ALU::%d] rstn = %b, control_bus = %b, opt1 = %d, opt2 = %d,
result = %d, stk_data_in = %d, stk_push = %b, stk_pop = %b, stk_data_out = %d z_flag = %b,

```



```
s_flag = %b,", state, rstn, control_bus, op1, op2, result, stk_data_in, stk_push, stk_pop,
stk_data_out, z_flag, s_flag);

endmodule
```

این ماژول برای دستورهای ADD/SUB بکار میرود. که برای این دستورات، نیاز داریم به POP/PUSH روی استک داریم.

یک استیت INIT داریم که در آن سیگنالهای اولیه را ست میکنیم و کار شروع میشود.

برای عملیات جمع و تفریق نیاز است دو تا عدد از استک POP کنیم.

کل عملیات POP کردن 3 حالت مختلف دارد. ابتدا سیگنال درخواست POP کردن فعال میشود. و بعد باید یک کلاک صبر کنیم و بعد عدد را از اول استک بیرون بیاوریم. و سپس در کلاک بعدی این عدد را در operand مان بریزیم. پس کلا 3 کلاک طول میکشد. پس برای POP کردن دو عدد 6 مرحله داریم که هرکدام یک کلاک طول میکشد. پس از اینکه هر دو عدد دریافت شد، در استیت ALU عملیات را انجام میدهیم. حال خروجی نهایی را نیز در استک باید ذخیره کنیم. که این عملیات PUSH کردن هم نیازمند 3 کلاک است. ابتدا سیگنالها فعال میشوند. سپس یک کلاک صبر میکنیم تا این سیگنال اعمال شود و PUSH انجام شود. یک مرحله سوم هم نیاز داریم که کارمان تمام شود و مثلاً POP یا PUSH بعدی را شروع کنیم. که این مرحله می تواند همان استیت INIT باشد. بنابراین پس از PUSH\_W مستقیم به INIT میرویم.

ماژول PC:

```
`define INIT      3'b000
`define POP       3'b001
`define NXTL      3'b010
`define BR        3'b011
`define EXIT      3'b100
`define POP_W     3'b101

`define OP_JP     2'b01
`define OP_JS     2'b10
`define OP_JZ     2'b11

module PC (control_bus,
```

```

    clk,
    rstn,
    en,
    pc,
    z_flag,
    s_flag,
    stk_pop,
    stk_push,
    stk_data_out,
    fin_sig);

parameter INST_CAP = 20;
parameter DATA_LEN = 8;

input clk, en, rstn, z_flag, s_flag;
input [3:0] control_bus;
input [DATA_LEN-1:0] stk_data_out;
output reg stk_pop,stk_push, fin_sig;
output reg [$clog2(INST_CAP):0] pc;

wire[1:0] opc = {control_bus[2], control_bus[0]};
wire branch  = (control_bus > 2 && control_bus < 6);
wire exit    = (control_bus == 4'b1111);
wire stall   = !(control_bus < 4'b1000 || control_bus == 4'b1111);

reg [2:0]state;

always @(posedge clk, negedge rstn) begin
    if (!rstn && !en) begin
        state    <= `INIT;
        stk_pop   <= 1'bz;
        stk_push  <= 1'bz;
        pc        <= 0;
        fin_sig   <= 0;
    end
    else begin
        case(state)
            `INIT:
                begin
                    if (en) begin
                        if (branch)
                            state <= `POP;
                        else if (exit)
                            state <= `EXIT;
                        else
                            state <= `NXTL;
                    end
                    fin_sig <= 0;
                    stk_pop <= 1'bz;
                    stk_push <= 1'bz;
                end
            `POP:

```

```

        begin
            stk_push <= 0;
            if ((opc == `OP_JP) || (opc == `OP_JZ && z_flag) || (opc == `OP_JS &&
s_flag)) begin
                stk_pop <= 1;
                state <= `POP_W;
            end
            else begin
                state <= `NXTL;
            end
        end
        `POP_W:
        begin
            stk_pop <= 1;
            state <= `BR;
        end
        `NXTL:
        begin
            if (pc < INST_CAP - 1) begin
                pc <= pc + 1;
            end
            state <= `INIT;
            if (stall) begin
                fin_sig <= 1;
            end
        end
        `BR:
        begin
            pc <= stk_data_out;
            state <= `INIT;
            fin_sig <= 1;
        end
        `EXIT:
        begin
            $writememb("report/result.mem", cpu0.memory0.mem);
            $finish;
        end
    endcase
end

end

//debugging
always @(*)
    $display($time, "\t [PC::%d] rstn = %b, en = %b, control_bus = %b, pc = %d, z_flag =
%b, s_flag = %b, stk_pop = %b, stk_data_out = %d", state, rstn, en, control_bus, pc, z_flag,
s_flag, stk_pop, stk_data_out);

endmodule

```

کلا ماژول PC سه حالت دارد. یا می خواهد برنامه را تمام کند و شروع به نوشتن اطلاعات جدید در حافظه کند، یا می خواهد دستور Jump ای را اجرا کند. و یا صرفاً می خواهد Program Counter را یک واحد افزایش دهد و برود خط بعدی را بخواند و الی آخر.

برای حالت اول که صرفاً کافیسٹ \$writememb را فراخوانی کنیم. برای حالت دوم، لازم است ابتدا یک بار از استک POP کنیم و بدانیم که به کدام خانه می خواهیم برویم. و سپس عملیات Jump را انجام دهیم و PC را برابر مقدار مربوطه ست کنیم. که این POP کردن و انجام عملیات، مشابه بخش ALU، در 3 مرحله مختلف انجام میشود. POP, POP\_W, BR. برای حالتی که صرفاً می خواهیم PC را یک واحد افزایش دهیم، کارمان راحت است. در حالت NXTL چک میکنیم که آیا به انتهای ظرفیت دستورات نرسیده باشیم، و سپس pc را یک واحد افزایش میدهیم. همچنین، اگر دستور قبلی stall بوده باشد، سیگنال fin\_sig که نشان دهنده پایان عملیات اجرای دستور است را = 1 می کنیم.

ماژول PP:

```
`define INIT      3'b000
`define LOAD      3'b011
`define PUSH      3'b001
`define PUSH_W    3'b010

`define POP       3'b101
`define POP_W     3'b110
`define STORE     3'b111

`define OP_PUSHC   2'b00
`define OP_PUSHM   2'b01
`define OP_POPM    2'b10

module PP (control_bus,
           clk,
           en,
           rstn,
```

```

    addr_const,
    stk_data_in,
    stk_push,
    stk_pop,
    stk_data_out,
    mem_data_in,
    mem_addr,
    mem_r_en,
    mem_w_en,
    mem_data_out,
    fin_sig);

parameter ADDR_LEN = 8;
parameter DATA_LEN = 8;

input clk, rstn, en;
input [3:0] control_bus;
input [DATA_LEN-1:0] stk_data_out, mem_data_out, addr_const;
output reg [DATA_LEN-1:0] stk_data_in, mem_data_in;
output reg [ADDR_LEN-1:0] mem_addr;
output reg stk_push, stk_pop, mem_r_en, mem_w_en, fin_sig;

wire [1:0] opc = control_bus[1:0];
wire pp_en    = (en && control_bus < 3);

reg [2:0] state;

always @(posedge clk, negedge rstn) begin
    if (!rstn && !pp_en) begin
        state      <= `INIT;
        stk_data_in <= {DATA_LEN{1'bz}};
        stk_push    <= 1'bz;
        stk_pop     <= 1'bz;
        mem_data_in <= {DATA_LEN{1'bz}};
        mem_addr    <= {ADDR_LEN{1'bz}};
        mem_r_en    <= 1'bz;
        mem_w_en    <= 1'bz;
        fin_sig     <= 0;
    end
    else begin
        case(state)
            `INIT:
                begin
                    if (pp_en) begin
                        state <= {opc, 1'b1};
                    end
                    fin_sig    <= 0;
                    stk_data_in <= {DATA_LEN{1'bz}};
                    stk_push    <= 1'bz;
                    stk_pop     <= 1'bz;
                    mem_data_in <= {DATA_LEN{1'bz}};
                    mem_addr    <= {ADDR_LEN{1'bz}};
                end
        endcase
    end
end

```

```

        mem_r_en    <= 1'bz;
        mem_w_en    <= 1'bz;
    end
    `PUSH:
    begin
        mem_r_en    <= 0;
        mem_w_en    <= 0;
        stk_pop     <= 0;
        stk_data_in <= (opc == `OP_PUSHC) ? addr_const : mem_data_out;
        state       <= `PUSH_W;
    end
    `PUSH_W:
    begin
        stk_push <= 1;
        state   <= `INIT;
        fin_sig <= 1;
    end
    `STORE:
    begin
        mem_w_en    <= 1;
        mem_r_en    <= 0;
        mem_data_in <= stk_data_out;
        mem_addr    <= addr_const;
        state       <= `INIT;
        fin_sig     <= 1;
    end
    `POP:
    begin
        stk_push <= 0;
        stk_pop  <= 1;
        state    <= `POP_W;
    end
    `POP_W:
    begin
        state <= `STORE;
    end
    `LOAD:
    begin
        mem_r_en <= 1;
        mem_w_en <= 0;
        mem_addr <= addr_const;
        state    <= `PUSH;
    end
    default:
        state <= `INIT;
    endcase
end

end

//debugging
always @(*)

```

```

    $display($time, "\t [WBPB:%d] rstn = %b, en = %b, fin_sig = %b, control_bus = %b,
    addr_const = %d, stk_data_in = %d, stk_push = %b, stk_pop = %b, stk_data_out = %d,
    mem_data_in = %d, mem_addr = %d, mem_r_en = %b, mem_w_en = %b, mem_data_out = %d", state,
    rstn, en, fin_sig, control_bus, addr_const, stk_data_in, stk_push, stk_pop, stk_data_out,
    mem_data_in, mem_addr, mem_r_en, mem_w_en, mem_data_out);

endmodule

```

در این ماژول، وظایف PUSH, POP, LOAD, STORE را داریم.

کلا سه نوع دستور داریم که به PP محول میشود:

PUSHC: push c

PUSHM: load m, push m

POPM: pop m, store m

در ابتدا در حالت INIT هستیم. اگر  $enable = 1$  شود، باید وارد state ای بشویم که وابسته به نوع دستورمان است. اگر از نوع PUSHC باشیم، همان ابتدا مستقیماً وارد فاز push میشویم و در 2 کلاک و در حالت های PUSH, PUSH\_W عملیات را انجام میدهیم. اگر دستورمان از نوع PUSH\_M باشد، ابتدا باید LOAD کنیم. و از حافظه بخوانیم. سپس، مشابه نوع قبلی وارد فاز PUSH و سپس PUSH\_W میشویم. اما اگر دستورمان POPM بود، لازم است ابتدا یک بار از استک POP کنیم، و سپس داده مورد نظر را در حافظه ذخیره کنیم. که این کار را در فاز STORE انجام میدهیم. پس در نهایت، استیت های پایانی ما، وابسته به اینکه نوع دستورمان چه باشد، یکی از دو استیت STORE و PUSH\_W است. پس در این دو حالت پس از اینکه کارها انجام شد، باید به استیت اولیه INIT برگردیم.

بقیه ماژول های مورد استفاده:

Stack:

```

module Stack (rstn,
              data_in,
              push,
              pop,
              clk,
              data_out,

```

```

        full,
        empty);

parameter STACK_DEPTH = 8;
parameter WORD_LEN     = 8;

input wire rstn, push, pop, clk;
input wire[WORD_LEN-1:0] data_in;
output full, empty;
output reg [WORD_LEN-1:0] data_out;

reg[$clog2(STACK_DEPTH):0] stack_ptr;
reg[WORD_LEN-1:0] memory [0:STACK_DEPTH-1];

assign empty = (stack_ptr == 0) ? 1'b1 : 1'b0;
assign full  = (stack_ptr == STACK_DEPTH) ? 1'b1 : 1'b0;

wire [WORD_LEN-1:0]top = memory[stack_ptr-1];

// stack reset
task reset_memory;
    integer i;
    begin
        for (i = 0; i < STACK_DEPTH;i++) begin
            memory[i] <= {WORD_LEN{1'b0}};
        end
        stack_ptr <= 0;
    end
endtask

always @(posedge clk or negedge rstn) begin
    if (!rstn) begin
        reset_memory;
    end
    else begin
        // pushing into stack
        if (push && !pop && !full) begin
            memory[stack_ptr] <= data_in;
            stack_ptr      <= stack_ptr + 1;
        end
        // pop from stack
        if (pop && !push && !empty) begin
            data_out <= memory[stack_ptr - 1];
            stack_ptr <= stack_ptr - 1;
        end
    end

end

//debugging
always @(*)
    $display($time, "\t [STACK] rstn = %b, stack_ptr = %d, top = %d, data_in = %d, push

```



```

= %b, pop = %b, data_out = %d, full = %b, empty = %b", rstn, stack_ptr, top, data_in, push,
pop, data_out, full, empty);

endmodule

```

## CPU

```

`define IFIDC    2'b00
`define IFIDC_W  2'b01
`define EXEC     2'b10
`define EXEC_W   2'b11

module CPU (rstn,
            clk);

    parameter DATA_LEN    = 8;
    parameter ADDR_LEN     = 8;
    parameter INST_CAP     = 20;
    parameter INST_LEN     = 12;
    parameter WORD_LEN     = 8;
    parameter MEM_SIZE     = 256;
    parameter STACK_DEPTH = 8;

    input rstn, clk;

    wire stk_full, stk_empty, stk_push, stk_pop, mem_r_en, mem_w_en, exec_fin_sig, IS_ready;
    wire [3:0] ifidc_control_bus;
    wire [DATA_LEN-1:0] stk_data_in, stk_data_out, mem_data_in, mem_data_out,
ifidc_addr_const;
    wire [ADDR_LEN-1:0] mem_addr;
    wire [$clog2(INST_CAP):0] exec_pc;

    reg ifidc_fetch, exec_en;
    reg [1:0] state;

    always @(posedge clk or negedge rstn) begin
        if (!rstn) begin
            ifidc_fetch <= 0;
            exec_en      <= 0;
            state        <= `IFIDC;
        end
        else begin
            case(state)
                `IFIDC:
                    begin
                        ifidc_fetch <= 1;
                        state        <= `IFIDC_W;
                    end
            end
        end
    end

```

```

        `IFIDC_W:
        begin
            ifidc_fetch <= 0;
            if (IS_ready) begin
                state <= `EXEC;
            end
        end
        `EXEC:
        begin
            exec_en <= 1;
            state <= `EXEC_W;
        end
        `EXEC_W:
        begin
            exec_en <= 0;
            if (exec_fin_sig) begin
                state <= `IFIDC;
            end
        end
    endcase
end
end

```

```

IFIDC #(
    .INST_CAP(INST_CAP),
    .INST_LEN(INST_LEN),
    .DATA_LEN(DATA_LEN)
) ifidc0 (
    .clk(clk),
    .en(ifidc_fetch),
    .rstn(rstn),
    .pc(exec_pc),
    .IS_ready(IS_ready),
    .control_bus(ifidc_control_bus),
    .data(ifidc_addr_const)
);

```

```

EXEC #(
    .DATA_LEN(DATA_LEN),
    .ADDR_LEN(ADDR_LEN),
    .INST_CAP(INST_CAP)
) exec0 (
    .clk(clk),
    .en(exec_en),
    .rstn(rstn),
    .pc(exec_pc),
    .control_bus(ifidc_control_bus),
    .addr_const(ifidc_addr_const),
    .stk_data_in(stk_data_in),
    .stk_push(stk_push),
    .stk_pop(stk_pop),
    .stk_data_out(stk_data_out),

```

```

.mem_data_in(mem_data_in),
.mem_addr(mem_addr),
.mem_r_en(mem_r_en),
.mem_w_en(mem_w_en),
.mem_data_out(mem_data_out),
.fin_sig(exec_fin_sig)
);

MEMORY #(
.ADDR_LEN(ADDR_LEN),
.WORD_LEN(WORD_LEN),
.MEM_SIZE(MEM_SIZE)
) memory0 (
.clk(clk),
.addr(mem_addr),
.r_en(mem_r_en),
.w_en(mem_w_en),
.data_out(mem_data_out),
.data_in(mem_data_in)
);

Stack #(
.WORD_LEN(WORD_LEN),
.STACK_DEPTH(STACK_DEPTH)
) stack0 (
.clk(clk),
.rstn(rstn),
.data_in(stk_data_in),
.push(stk_push),
.pop(stk_pop),
.data_out(stk_data_out),
.full(stk_full),
.empty(stk_empty)
);

//debugging
always @(*)
    $display($time, "\t [CPU:%d] ifidc_fetch = %b, exec_en = %b, exec_finish = %b",
state, ifidc_fetch, exec_en, exec_fin_sig);

endmodule

```

## Testbench

```

module testbench();

parameter CLK_C      = 10;
parameter DATA_LEN  = 8;

```

```

parameter ADDR_LEN      = 8;
parameter INST_CAP      = 20;
parameter INST_LEN      = 12;
parameter WORD_LEN      = 8;
parameter MEM_SIZE      = 256;
parameter STACK_DEPTH = 8;

reg clk, rstn;

CPU #(
    .WORD_LEN(WORD_LEN),
    .DATA_LEN(DATA_LEN),
    .ADDR_LEN(ADDR_LEN),
    .MEM_SIZE(MEM_SIZE),
    .STACK_DEPTH(STACK_DEPTH),
    .INST_LEN(INST_LEN),
    .INST_CAP(INST_CAP)
) cpu0 (
    .clk(clk),
    .rstn(rstn)
);

initial begin
    $dumpfile("report/waveform.vcd");
    $dumpvars(0,cpu0);
end

initial begin
    clk      = 0;
    forever clk = #(CLK_C/2) ~clk;
end

initial begin
    $readmemb("report/memory.mem", cpu0.memory0.mem, 0, MEM_SIZE-1);
    $readmemb("report/is.mem", cpu0.ifidc0.inst_mem, 0, INST_CAP-1);
end

initial begin
    rstn = 0;
    #CLK_C
    rstn = 1;
    #(100000 * CLK_C);
    $finish;
end

endmodule

```

## ls\_math.mem

```
// Example: Y=((X+23)+(X+23))-12
0001_0000_0001      // PUSHM X
0000_0001_0111      // PUSHC 23
0110_xxxx_xxxx      // ADD
0010_0000_0010      // POPM Y
0001_0000_0010      // PUSHM Y
0001_0000_0010      // PUSHM Y
0110_xxxx_xxxx      // ADD
0000_0000_1100      // PUSHC 12
0111_xxxx_xxxx      // SUB
0010_0000_0010      // POPM RES
1111_xxxx_xxxx      // EXIT
0000_0000_0000
0000_0000_0000
0000_0000_0000
0000_0000_0000
0000_0000_0000
0000_0000_0000
0000_0000_0000
0000_0000_0000
0000_0000_0000
```

## ls.mem

```
// Example: Y=((X+23)+(X+23))-12
0001_0000_0001      // PUSHM X
0000_0001_0111      // PUSHC 23
0110_xxxx_xxxx      // ADD
0010_0000_0010      // POPM Y
0001_0000_0010      // PUSHM Y
0001_0000_0010      // PUSHM Y
0110_xxxx_xxxx      // ADD
0000_0000_1100      // PUSHC 12
0111_xxxx_xxxx      // SUB
0010_0000_0010      // POPM RES
1111_xxxx_xxxx      // EXIT
0000_0000_0000
0000_0000_0000
0000_0000_0000
0000_0000_0000
0000_0000_0000
0000_0000_0000
0000_0000_0000
0000_0000_0000
0000_0000_0000
```

## Memory.mem

```
// main memory: 256 * word, word = 8 bits
0000_0000
0000_0101
0000_1010
0000_0000
0000_0000
0000_0000
0000_0000
0000_0000
0000_0000
0000_0000
0000_0000
0000_0000
0000_0000
0000_0000
0000_0000
0000_0000
0000_0000
0000_0000
0000_0000
0000_0000
```

## Result.mem

```
// 0x00000000
00000000
00000101
00101100
00000000
00000000
00000000
00000000
00000000
00000000
00000000
00000000
00000000
00000000
00000000
00000000
00000000
00000000
```

## Is\_jump.mem

```
// Branch instruction tests
```

```

0000_0000_0011      // 0.PUSHC 3
0000_0000_0011      // 1.PUSHC 3
0111_xxxx_xxxx      // 2.SUB
0000_0000_0110      // 3.PUSHC 6
0100_xxxx_xxxx      // 4.JZ
1111_xxxx_xxxx      // 5.EXIT
0000_0000_1000      // 6.PUSHC 8
0000_0001_0000      // 7.PUSHC 16
0111_xxxx_xxxx      // 8.SUB
0000_0000_1100      // 9.PUSHC 12
0101_xxxx_xxxx      // 10.JS
1111_xxxx_xxxx      // 11.EXIT
0000_0000_0101      // 12.PUSHC 5
0011_xxxx_xxxx      // 13.JUMP
0000_0000_0000
0000_0000_0000
0000_0000_0000
0000_0000_0000
0000_0000_0000
0000_0000_0000

```

## ls\_swap.mem

```

// SWAP
0001_0000_0001      // PUSHM ARG1
0001_0000_0010      // PUSHM ARG2
0010_0000_0001      // POPM ARG2
0010_0000_0010      // POPM ARG1
1111_xxxx_xxxx      // EXIT
0000_0000_0000
0000_0000_0000
0000_0000_0000
0000_0000_0000
0000_0000_0000
0000_0000_0000
0000_0000_0000
0000_0000_0000
0000_0000_0000
0000_0000_0000
0000_0000_0000
0000_0000_0000
0000_0000_0000
0000_0000_0000
0000_0000_0000
0000_0000_0000
0000_0000_0000
0000_0000_0000

```

## Results.txt

[illegible]




25

WARNING: PC.v:98: \$writememb: Standard inconsistency, f

شکل موجها به شرح زیر است:


## IFIDC\_EXEC




## ALU




## Memory



## PP\_PC



## Stack





پایان (: